
Subject: Re: Force Field

Posted by [David Fanning](#) on Tue, 13 Sep 2005 18:22:41 GMT

[View Forum Message](#) <> [Reply to Message](#)

Andres writes:

- > I have a potential field on a grid (it has periodic boundary
- > conditions), and I want to calculate the acceleration (force field) of
- > this, i.e., $f = -\text{grad}(\text{potential})$, I need the gradient of the potential .
- > I wrote a procedure, but it has too many for loops in it. Does any body
- > know a fast and smart way to do this?

Here is a function I use to calculate the gradient of a 2D image:

FUNCTION Gradient, image

```
    ; gradient ~= |Gx| + |Gy|
```

```
    COMPILE_OPT idl2
```

```
    ; Check parameters.
```

```
    ON_ERROR, 2
```

```
    IF N_Elements(image) EQ 0 THEN Message, 'Must pass 2D image data.'
```

```
    ; Calculate gradient Gx.
```

```
    xkernel = [ [-1.0, 0.0, 1.0], [-2.0, 0.0, 2.0], [-1.0, 0.0, 1.0] ]
```

```
    gx = Convol( Float(image), xkernel, Center=1, /Edge_Wrap )
```

```
    ; Calculate gradient Gy.
```

```
    ykernel = [ [-1.0, -2.0, -1.0], [ 0.0, 0.0, 0.0], [ 1.0, 2.0, 1.0] ]
```

```
    gy = Convol( Float(image), ykernel, Center=1, /Edge_Wrap )
```

```
    RETURN, ABS(gx) + ABS(gy)
```

```
END
```

I use it like this:

```
IDL> TVSCL, Gradient(image2d)
```

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.dfanning.com/>
